<commit_before>{ stdenv, fetchFromGitHub, libpcap, sqlite, pixiewps, makeWrapper }:

stdenv.mkDerivation rec {
  version = "1.5.2";
  name = "reaver-wps-t6x-${version}";
  confdir = "/var/db/${name}"; # the sqlite database is at "${confdir}/reaver/reaver.db"

  src = fetchFromGitHub {
    owner = "t6x";
    repo = "reaver-wps-fork-t6x";
    rev = "v${version}";
    sha256 = "0zhlms89ncqz1f1hc22yw9x1s837yv76f1zcjizhgn5h7vp17j4b";
  };

  nativeBuildInputs = [ makeWrapper ];
  buildInputs = [ libpcap sqlite pixiewps ];

  setSourceRoot = ''
    sourceRoot=$(echo */src)
  '';

  configureFlags = "--sysconfdir=${confdir}";

  installPhase = ''
    mkdir -p $out/{bin,etc}
    cp reaver.db $out/etc/
    cp reaver wash $out/bin/

    wrapProgram $out/bin/reaver --run "[ -s ${confdir}/reaver/reaver.db ] || install -D $out/etc/reaver.db ${confdir}/reaver/reaver.db"
    wrapProgram $out/bin/wash   --run "[ -s ${confdir}/reaver/reaver.db ] || install -D $out/etc/reaver.db ${confdir}/reaver/reaver.db"
  '';

  meta = with stdenv.lib; {
    description = "Online and offline brute force attack against WPS";
    homepage = https://github.com/t6x/reaver-wps-fork-t6x;
    license = licenses.gpl2Plus;
    platforms = platforms.linux;
    maintainers = with maintainers; [ nico202 volth ];
  };
}
<commit_msg>reaver-wps-t6x: 1.5.2 -> 1.6.3<commit_after>{ stdenv, fetchFromGitHub, libpcap, pixiewps, makeWrapper }:

stdenv.mkDerivation rec {
  version = "1.6.3";
  name = "reaver-wps-t6x-${version}";

  src = fetchFromGitHub {
    owner = "t6x";
    repo = "reaver-wps-fork-t6x";
    rev = "v${version}";
    sha256 = "1bccwp67q1q0h5m38gqxn9imq5rb75jbmv7fjr2n38v10jcga2pb";
  };

  nativeBuildInputs = [ makeWrapper ];
  buildInputs = [ libpcap pixiewps ];

  preConfigure = "cd src";

  installPhase = ''
    mkdir -p $out/bin
    cp reaver wash $out/bin/
  '';

  meta = with stdenv.lib; {
    description = "Online and offline brute force attack against WPS";
    homepage = https://github.com/t6x/reaver-wps-fork-t6x;
    license = licenses.gpl2Plus;
    platforms = platforms.linux;
    maintainers = with maintainers; [ nico202 volth ];
  };
}

{'fraction_non_alphanumeric': 0.15527661552766156, 'fraction_numerical': 0.03533240353324035, 'mean_word_length': 3.2954091816367264, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 4, 'https://': 2, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '7582426', 'n_tokens_mistral': 901, 'n_tokens_neox': 850, 'n_words': 157}